

## MS 365 Excel Basics 07: What You Must Know To Build Efficient Formula Solutions

<b>What you must know to create efficient Excel worksheet formula solutions:</b>	<b>2</b>
1) Starting a formula	2
2) What Formulas do	2
3) Keyboard shortcuts help create formulas and formula models quickly	2
4) Default alignment in the Excel worksheet	2
5) The importance of default alignment	2
6) Number Formatting	3
7) Excel's Golden Rule	3
8) Formula elements (stuff you can put in a formula)	4
9) When to use Arrow Keys to enter cell references into a formula	6
10) Cell References	6
11) Example of Absolute and Relative Cell References	7
12) Example a Mixed Cell Reference and Conditional Formatting a Row in a Table	7
13) Operators used in Excel worksheet formulas	8
14) TEXTAFTER and TEXTBEFORE Functions	8
15) TEXTJOIN Function	8
16) Order of Operations for worksheet formulas	9
17) Evaluate Formula feature	9
18) Types of formulas	10
19) Examples formulas by data type	11
20) Convert text numbers to numbers	11
21) Examples of types of formulas by calculation type	12
22) Excel Table feature	13
23) Dynamic Spilled Array Formula to Build Budget	14
24) Rounding numbers.	15
25) Date and Time Number Formatting and formulas	17
26) Modes of cell editing	18
27) Logical tests in Excel worksheet formulas	19
28) Lookup functions	20
29) Formula Errors	21

## What you must know to create efficient Excel worksheet formula solutions:

- 1) Starting a formula
  - i. All formulas have an equal sign (=) as first character in cell.
- 2) What Formulas do
  - i. Excel worksheet formulas perform math, logical or data manipulation operation to create an answer, and they are the only feature in Excel that updates instantly when source data changes.
- 3) Keyboard shortcuts help create formulas and formula models quickly
  - i. Alt += to invoke the SUM function
  - ii. F4 to add and toggle through the different types of cell references
  - iii. Ctrl + Shift + ↓ to select a range in a column.
- 4) Default alignment in the Excel worksheet
  - i. Number (right)
  - ii. Text (left)
  - iii. Logical (centered & capitalized).
- 5) The importance of default alignment
  - i. It is important to instantly recognize default alignment to help detect errors, such as text numbers that cannot be summed with the SUM function (as shown below).

A	B	C	D	E	F	G	H	I
2	<b>Default Alignment Indicates Data Type:</b>							
3	Numbers aligned to right	43						
4	Text aligned to left	Excel						
5	Logicals are centered & capitalized	TRUE						
6								
7	<b>Can you name all data types in the range B8:I14?</b>							
8						<b>Imported From Database</b>		
9						541.95		
10						127.01		
11	<b>Bonus (\$)</b>	500				719.34		
12	<b>Bonus Hurdle (Equal to or Bigger) (\$)</b>	85,000				95.41		
13	<b>Your Sales</b>	78,500				339.27		
14	<b>Get bonus?</b>	FALSE	=C13>=C12		<b>Total</b>	0	=SUM(G9:G13)	
15								
16	<b>Answer:</b>					<b>Text</b>		
17						Text		
18						Text		
19	<b>Text</b>	Number				Text		
20	<b>Text</b>	Number				Text		
21	<b>Text</b>	Number				Text		
22	<b>Text</b>	Logical	Text		Text	Number	Text	
23								
24	<b>Default alignment is a visual que about whether you are entering data correctly:</b>							
25								
26	<b>Required</b>	<b>Typing error</b>						
27	Date	1/32/2024						
28	Sales	43..69						
29	Logical	treu						
30								
31	<b>Required</b>	<b>No typing error</b>						
32	Date	1/31/2024						
33	Sales	43.69						
34	Logical	TRUE						

## 6) Number Formatting

- i. Number Formatting displays numbers in a certain way without changing the underlying number.
- ii. General Number Format wipes away all Number Formatting. Keyboard: Ctrl + Shift + ~
- iii. Formulas do not "see" Number Formatting. Formulas act on the underlying number.

**Actual number in cell** ← 43.9

	A	B	C	D	E	F	G
1							
2		Numbers					
3			44				
4			43.9				
5			43.90				
6							
7		Multiple	2				
8		2*44 =	87.8		=B3*C7		
9							
10	Number format makes the number 43.9 in cell B3 display as 44, but the underlying number of 43.9 is the number that the formula uses to make the calculation.						
11	Formulas do not "see" or act on Number Formatting.						

**Display** ← 44

---

**Actual percent in formula bar** ← 5.5%

	A	B	C	D	E	F	G
12							
13		Sales	1,000				
14		Commission Rate	6%				
15		Commission Paid	55		=C14*C13		
16							
17	Number format makes the number 0.055 in cell C14 display as 6%, but the underlying number of 0.055 is the number that the formula uses to make the calculation.						
18	Formulas do not "see" or act on Number Formatting.						

**Display** ← 6%

## 7) Excel's Golden Rule

- i. If a formula input can change, put it in a cell, label it, and refer to it in the formula with a cell reference.

	A	B	C	D	E	F	G	H
2		<b>Excel's Golden Rule:</b>						
3		If a formula input can change, put it in a cell, label it, and refer to it in the formula with a cell reference.						
4								
5		<b>Example of formula inputs that can change:</b>			<b>Example of formula inputs that will never change:</b>			
6		Tax Rate	0.0765		Hours in day	24		
7		Product	Quad		Months in year	12		
8		Sales	500.35		Original Principal in Finance Form	1		
9								
10		<b>Goal:</b> Calculate the tax to be paid if the sales amount is \$100 and the tax rate is 5%						
11								
12		Sales amount (\$)	100					
13		Tax Rate	5.00%					
14		Tax To Be Paid (\$)	5	=C12*C13				
15								
16		<b>Goal:</b> If the sales for your company this year are \$230,000, and this is 15% higher than last year's sales, what were last year's sales?						
17		You can use the formula: <b>Begin = End/(1+%Change)</b>						
18					100 + 100*10% = 110			
19		Sales This Year (\$)	230,000.00		100*(1+10%) = 110			
20		% Increase in sales this year over the sales from last year	15.00%		110/(1+10%) = 100			
21		Sales Last Year (\$)	200,000.00	=C19/(1+C20)	End/(1+%Increase) = Begin			
22		Check Your Answer:	230,000.00	=C21*(1+C20)				

8) Formula elements (stuff you can put in a formula)

Element	Description	Example
Equal Sign	As first character in cell, it tells Excel you are making a formula	↓ =SUM(I2:I6)
Cell References	Four types of cell references: Relative, Absolute, Mixed with Row Locked, Mixed with Column Locked	G2, \$G\$2, G\$2, \$G2
Range of Cells	Four types of ranges: Relative, Absolute, Mixed with Row Locked, Mixed with Column Locked	G2:G5, \$G\$2:\$G\$5, G\$2:G\$5, \$G2:\$G5
Reference Operator: colon	Used in between two cell reference, two sheet names or two lookup functions to create a range of cells	↓ ↓ ↓ G2:G5, Jan:Apr!C3 XLOOKUP(V2,V3:Z3,V4:Z4):XLOOKUP(W2,V3:Z3,V4:Z4)
Reference Operator: comma	Used to combine ranges. Works in FREQUENCY, IRR, INDEX, AREAS, LET and aggregate functions like SUM and LARGE. Works in Defined Names.	↓ LARGE((V5:Z5,Sales),3)
Reference Operator: space	Used as an intersection operator to get the value at the intersection of two ranges	↓ G2:G6 F4:H4 yields "Quad"
Spilled Range Operator (#)	Spilled Range Operator (#) that allows you to refer to all items that are spilled from a cell with a spilled array formula	↓ C3#
Implicit Intersection Operator (@)	Implicit Intersection Operator (@) allows you to get a corresponding item in the current row of a column or a parallel column. Most commonly seen in relative cell reference in Excel Table columns.	↓ ↓ ↓ fSales[@Product], @Product, @G2:G6
Data Type Operator (Dot Operator)	For cells that contains a record, the dot allows you to specify what field you want to extra from the cell	↓ ↓ [@Stock].Price . B12.Price
Worksheet Reference	Name of worksheet (in single quotes if a space in name) with exclamation point before cell reference. Any one of four cell references allowed.	LookupTable!B3:B6 , 'Lookup Table'!B3:B6
Workbook Reference	Same as Worksheet Reference with name of workbook in square brackets and full file path. Any one of four cell references allowed.	<b>If referenced workbook opened:</b> '[Ch12-Excel365-WorksheetFormulas.xlsx]Elements'!\$G\$5, <b>If referenced workbook NOT opened:</b> 'F:\[Ch12-Excel365-WorksheetFormulas.xlsx]Elements'!G5
Defined Names	Defined Names are names that can represent a cell, ranges of cells, or formulas. The Defined Names can be used in other formulas and features, including in Power Query and the Power Pivot Data Model.	Defined Name to define a cell: <b>CallRepConditionAnswer</b> ='Ch12(31-34an)'!\$G\$9  Defined Name to define formula: <b>LookupPicture</b> =XLOOKUP('Ch14(38-40)'!\$B\$6,'Ch14(38-40)'!\$B\$9:\$B\$12,'Ch14(38-40)'!\$D\$9:\$D\$12)

Element	Description	Example
Table Formula Nomenclature in an Excel Table that has name <b>fSales</b> , and columns: <b>Date</b> , <b>Product</b> and <b>Sales</b>	Full Table	CallTable[#All]
	Records in Table	CallTable
	Field Names	CallTable[#Headers]
	Mixed with Row Locked Column Reference	CallTable[Rep]
	Locked Column Reference	CallTable[[Rep]:[Rep]]
	Selected Columns	CallTable[[Rep]:[Calls]]
Worksheet Functions	Relative Cell Reference (Implicit Intersection Operator)	CallTable[@Rep]
	Built-in function like SUM and ROUND that are programmed to make a defined calculation based on the formula inputs entered into the function's arguments	SUM, COUNTIFS, SUMIFS, FORMULATEXT, ROUND, MROUND, UNIQUE, SORT, EDATE, EOMONTH, NETWORKDAYS.INTL, WORKDAY.INTL, AVERAGE, SEQUENCE, LARGE, RANK.EQ, XLOOKUP, FILTER and many
Function argument elements	Some functions require that you select an item from a dropdown list in order to instruct the function which type of calculation to make, like with NETWORKDAYS.INTL that needs a 7 to instruct it to ignore Fridays and Saturdays	NETWORKDAYS.INTL(F2,F2, <b>7</b> )
Math Operators	Used to create math calculations	+ Adding - Subtracting or Negation * Multiplying / Dividing ^ Raising to an exponent ( ) Parentheses
Comparative Operator	Used to create comparative calculations	= Equal: are two things equal? <> Not: are two things Not equal? (less & greater than symbol) > Greater than: left side greater than right side? >= Greater than or Equal to: left side >= than right side? < Less than: left side less than right side?
Join Operator (&)	Ampersand symbol (&) to join two items into one items	C8&","&B8 , "Item # "&C8:C12
Text within quotation marks	Text in Formulas must be in quotes	C8&","&B8 , "Item # "&C8:C12
Hard Coded Numbers	When a number will not change, like 12 months in a year or 24 hours in a day, you can hard code it into formula.	C10/ <b>12</b> , (EndTime-StartTime)* <b>24</b>
Array constants (hard coded arrays)	Hard coded tables, columns or rows. Curly Brackets house the array: { } Comma means column , Semi-colon means row ;	{"Jan","Feb","Mar","Apr","May";1,2,3,4,5}

- 9) When to use Arrow Keys to enter cell references into a formula
- If cell references are close to the formula, it can be fast to use the arrow keys.
  - If the cell references are far away from the formula, it maybe faster to use the mouse.

## 10) Cell References

- 1) Example of Cell Reference: A1
  - Column reference = A
  - Row reference = 1
- 2) Copying formulas with Cell References:
  - When we copy a formula that contains cell references, we need to consider whether we need: Relative, Absolute, Mixed with the Column Locked or Mixed with the Row Locked.
  - If you will not copy the formula, there is no need to consider what type of cell reference it will be.
- 3) Four Basic Types of Cell References (Relative, Absolute, Mixed Column Locked, Mixed Row Locked):
  - Relative Cell References – Example: A1
    - No dollar signs
    - Moves relatively throughout the copy action.
    - Relatively means that if the formula is looking at a cell reference that is three cells to the left, when you copy the formula to any other cell, the cell reference will still be looking three cells to the left.
  - Absolute Cell References – Example: \$A\$1
    - Dollar signs before both:
      - Column reference = A
      - Row reference = 1
    - Absolute means that if the formula is looking at a particular cell reference, when you copy the formula to any other cell, the cell reference will still be looking at that particular cell reference. If the absolute cell reference is \$A\$1, the formula will always look at cell A1. It is as if the formula is locked on the cell A1 throughout copy action.
  - Mixed Cell References with Row Locked – Example: A\$1
    - Dollar sign before row reference only.
    - Remains absolute or locked when copying across the rows, vertically (up and down).
    - Moves relatively when copying across the columns, horizontally (side to side).
  - Mixed Cell References with Column Locked – Example: \$A1
    - Dollar sign before column reference only.
    - Remains absolute or locked when copying across the columns, horizontally (side to side).
    - Moves relatively when copying across the rows, vertically (up and down).
- 4) Keyboard to Toggle Cell References = F4 Key.
  - F4 key = If cursor is touching a cell reference in a formula while in edit mode, F4 toggles between the four basic types of cell references.

11) Example of Absolute and Relative Cell References

	F	G	H	I	J	K	L	M	N	O	P
1											
2		<b>Goal #1:</b>				<b>Formula must be copied and it needs locked (absolute) and relative cell references:</b>					
3		List products that each employee sold last week				M6: =TEXTJOIN(", ",UNIQUE(SORT(FILTER(\$H\$6:\$H\$24,\$J\$6:\$J\$24=L6))))					
4											
5		<b>Date</b>	<b>Product</b>	<b>Sales</b>	<b>Sales Rep</b>	<b>Sales Rep</b>	<b>What Products Did They Sell Last Week?</b>				
6		11/5/2024	Aspen	818.25	Chantel	Abby	Aspen, Carlota				
7		11/7/2024	Sunset	597.85	Dean	Chantel	Aspen, Beaut, Quad, Sunshine				
8		11/7/2024	Sunshine	418.46	Chantel	Dean	Carlota, Sunset				
9		11/8/2024	Aspen	732.22	Max	Max	Aspen, Sunset				
10		11/7/2024	Carlota	726.71	Dean	Shelladawn	Beaut				
11		11/8/2024	Beaut	846.07	Chantel						
12		11/4/2024	Sunset	392.16	Dean						

12) Example a Mixed Cell Reference and Conditional Formatting a Row in a Table

	A	B	C	D	E	F	G	H	I	J	K	L
2		<b>Formula Goal #2:</b>										
3		Conditionally format each row with the designated product:										
4												
5		<b>Product</b>	Sunshine									
6												
7		<b>The only way to conditionally format a row in a table based on a condition in the record is to use a Mixed Cell Reference with Column Locked.</b>										
8												
9								Formula used in Conditional Format dialog box when the Active Cell is B13,				
10								and the formula is used in cell G12 in the worksheet:				
11								=\$C13=\$C\$5				
12		<b>Date</b>	<b>Product</b>	<b>Sales</b>	<b>Sales Rep</b>							
13		11/2/2024	Aspen	818.25	Chantel	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE
14		11/12/2024	Sunset	597.85	Dean	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE
15		11/4/2024	Sunshine	418.46	Chantel	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE
16		11/17/2024	Aspen	732.22	Max	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE
17		11/15/2024	Sunshine	789.2	Chantel	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE	TRUE
18		11/13/2024	Carlota	726.71	Dean	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE
19		11/5/2024	Beaut	846.07	Chantel	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE

### 13) Operators used in Excel worksheet formulas

Math Operators
( ) represents Parentheses
^ represents Exponents (powers and roots)
* represents Multiplication
/ represents Division
+ represents Addition
- represents Subtraction

Comparative Operators
> Greater Than
>= Greater Than Or Equal To
< Less Than
<= Less Than Or Equal To
= Equal To
<> Not Equal To

Join Symbol (Ampersand)
&

Reference Operators
: colon Used in between two reference, like with =SUM(A5:A10)
, comma Used to combine ranges, like with =SUM(A5:A10 , D5:D10)
# hash tag (Spilled Range Operator) indicates that all spilled values should be referenced, like: B6#
@ symbol (Implicit Intersection Operator) indicates that the value in the current row must be used, like: [@Sales]

### 14) TEXTAFTER and TEXTBEFORE Functions

- i. **TEXTAFTER** function can extract text after a delimiter  
=TEXTAFTER(text,delimiter,[instance\_num], [match\_mode], [match\_end], [if\_not\_found])
- ii. **TEXTBEFORE** function can extract text before a delimiter.  
=TEXTBEFORE(text,delimiter,[instance\_num], [match\_mode], [match\_end], [if\_not\_found])
- iii. Both functions have the same arguments:
  - text** The text you are searching within. Wildcard characters not allowed. Required.
  - delimiter** The text that marks the point after which you want to extract. Required.
  - instance\_num** The instance of the delimiter after which you want to extract the text. By default, instance\_num = 1. A negative number starts searching text from the end. Optional.
  - match\_mode** Determines whether the text search is case-sensitive. The default is case-sensitive. Optional. Enter one of the following: **0** Case sensitive, **1** Case insensitive.
  - match\_end** Treats the end of text as a delimiter. By default, the text is an exact match. Optional. Enter one of the following: **0** Don't match the delimiter against the end of the text, **1** Match the delimiter against the end of the text.
  - if\_not\_found** Value returned if no match is found. By default, #N/A is returned. Optional.

### 15) TEXTJOIN Function

- i. **TEXTJOIN** can join text with a delimiter between each text item.  
TEXTJOIN(delimiter, ignore\_empty, text1, [text2], ...)
- ii. Function arguments:
- iii. delimiter (required) A text string, either empty, or one or more characters enclosed by double quotes, or a reference to a valid text string. If a number is supplied, it will be treated as text.
- iv. ignore\_empty (required). If TRUE, ignores empty cells.
- v. text1 (required). Text item to be joined. A text string, or array of strings, such as a range of cells.
- vi. [text2, ...] (optional). Additional text items to be joined. There can be a maximum of 252 text arguments for the text items, including text1. Each can be a text string, or array of strings, such as a range of cells.




## 16) Order of Operations for worksheet formulas

A	B	C	D	E	F	G																		
2	<b>Excel Formula Order of Operations:</b>		<b>Example 1:</b>																					
3	<b>1) Parenthesis ( )</b>		<b>Must use ( ) to get - to happen before *</b>																					
4	<b>2) Reference Operators:</b> colon, space, comma, hash tag, @ symbol,		<table border="1"> <thead> <tr> <th>Time In</th> <th>Time Out</th> <th>Hours Worked</th> </tr> </thead> <tbody> <tr> <td>9:30 AM</td> <td>3:00 PM</td> <td>5.5</td> </tr> </tbody> </table>				Time In	Time Out	Hours Worked	9:30 AM	3:00 PM	5.5												
Time In	Time Out	Hours Worked																						
9:30 AM	3:00 PM	5.5																						
5	table formula nomenclature		=(F5-E5)*24																					
6	Example of colon in range of cells: =SUM(A1:A4)		<b>Example 2:</b>																					
7	Example of intersection operator: =E12:G12 F10:F15 (retrieve what is in F12)		<b>&lt; is evaluated after *, so no ( ) needed</b>																					
8	Example of comma (union): =SUM(E10:G10,E14:G14)		<table border="1"> <thead> <tr> <th>Price</th> <th>Units</th> <th>50&lt;Sales?</th> </tr> </thead> <tbody> <tr> <td>15</td> <td>4</td> <td>TRUE</td> </tr> </tbody> </table>				Price	Units	50<Sales?	15	4	TRUE												
Price	Units	50<Sales?																						
15	4	TRUE																						
9	Example of hash tab: =SUM(B6#)		=50<E10*F10																					
10	Example of Implicit Intersection Operator (@): =[Product]		<b>Example 3:</b>																					
11	Example of Table Formula Nomenclature: =SUM(Sales[Sales])		<b>Must use ( ) to get + to happen before &gt;</b>																					
12	<b>3) Negation (-)</b> (give me opposite)		<table border="1"> <thead> <tr> <th>Sales (\$K) &gt;</th> <th>CR &gt;</th> <th>OR Count</th> </tr> </thead> <tbody> <tr> <td>250</td> <td>4</td> <td>3</td> </tr> </tbody> </table>				Sales (\$K) >	CR >	OR Count	250	4	3												
Sales (\$K) >	CR >	OR Count																						
250	4	3																						
13	Example: = -2^4 = 16		=SUM(IF((F20:F24>E16)+(G20:G24>F16),1))																					
14	Example: = -(2^4) = -16		<table border="1"> <thead> <tr> <th>Customer</th> <th>Sales (\$K)</th> <th>Credit Rating</th> </tr> </thead> <tbody> <tr> <td>SW</td> <td>275</td> <td>5</td> </tr> <tr> <td>PCC</td> <td>225</td> <td>4.5</td> </tr> <tr> <td>T</td> <td>195</td> <td>3</td> </tr> <tr> <td>QFC</td> <td>115</td> <td>6</td> </tr> <tr> <td>WM</td> <td>95</td> <td>1.5</td> </tr> </tbody> </table>				Customer	Sales (\$K)	Credit Rating	SW	275	5	PCC	225	4.5	T	195	3	QFC	115	6	WM	95	1.5
Customer	Sales (\$K)	Credit Rating																						
SW	275	5																						
PCC	225	4.5																						
T	195	3																						
QFC	115	6																						
WM	95	1.5																						
15	Example: --2+1 = 3																							
16	<b>4) Converts %</b> (1% to .01)																							
17	<b>5) Exponents (^)</b>																							
18	Example: 3^2 = 9																							
19	Example: 2^3 = 2*2*2 = 8																							
20	Example: 4^(1/2) = 2																							
21	<b>6) Multiplication (*) and Division (/), left to right</b>																							
22	<b>7) Adding (+) and Subtracting (-), left to right</b>																							
23	<b>8) Ampersand (&amp;)</b> (Join operator)																							
24	<b>9) Comparative operators:</b> =, <>, >=, <=, <, >																							

## 17) Evaluate Formula feature

- You can step through how Excel calculates a formula step-by-step by selecting the cell with the formula and then in the Formula tab, Formula Auditing group, click the Evaluate Formula button, as shown below:



## 18) Types of formulas

### Formulas by data type:

**1) Number Formula:** Use Math Operators or Functions to calculate a number answer. Numbers are aligned to the right by default.

**2) Logical Formula:** Use a Comparative Operators or Functions to deliver a TRUE or FALSE. TRUE & FALSE are aligned in the Center by default.

**3) Text Formula:** Use Join Operator, "Text" in quotes or Functions (like LEFT, TEXT, FIXED and more) to deliver a text answer. Text is aligned to the left by default.

### Formulas by calculation type:

**1) Aggregate Formulas** operate on an array of values and deliver a single answer, like with adding, averaging or running an AND logical test.

**2) Single Input-Output Formulas** are formulas that operate on single inputs and deliver a single answer. This type of formula has single values entered into a function argument, like =FORMULATEXT(B1), or has single values on either side of an operator, like =A1\*B1, where the operator is a multiplication operator.

#### These formulas require that you:

- 1) Consider what type of cell references (relative, absolute, mixed) are required.
- 2) Enter the formula into one cell, and if the formula must be copied, you must manually copy the formula to other cells.
- 3) If you need to edit the formula, you edit the cell and re-copy the formula to other cells if necessary.

**Array Formulas** are formulas that contain one or more array operations that deliver an array of answers rather than a single answer. The array operation can involve multiple values entered into a function argument, like =FORMULATEXT(B1:B4), or there can be multiple values on one or more sides of an operator, like =A1\*B1:B5, or A1:A5\*B1, or A1:A5\*B1:B5. Array formula is the general term used to describe both types of array formulas:

**3) Dynamic Spilled Array Formulas** are array formulas that deliver a spilled array to the worksheet as the final answer. Dynamic Spilled Array formulas spill from the top cell and only live in the top cell. If you spill a formula from cell F9, you refer to the spilled array with the spilled range operator #, as in SUM(#F9) when you want to add the values.

**4) Scalar Array Formulas** are array formulas that deliver a single, scalar value as the final answer. For example: =SUM(ROUND(M9:M13\*N6,2)), where the array operation M9:M13\*N6 delivers an array of values to ROUND and then SUM adds the values to deliver a single scalar value (one value).

**5) Excel Table Formulas** are formulas that reference an Excel Table. The reason that we use Excel Table Formulas is that when new records are added to or removed from an Excel Table, all formulas and Excel features referencing the Excel Table will reflect the updated data.

**Table formula nomenclature** are references to Excel Table objects, such as:

TaxTable = Excel Table Name.

TaxTable[Tax Amount] = Refers to entire TaxAmount column in the Excel Table named TaxTable.

[@Sales] refers to the current row value in the Excel Table's Sales column. The @ operator is called Implicit Intersection Operator.

Examples of formulas that use table formula nomenclature:

=SUM(TaxTable[Tax Amount]) is an aggregate formula that adds all the values from the TaxAmount column in the Excel Table named TaxTable.

=ROUND([@Sales]\*\$16,2) is an Excel Table column formula that calculates the tax amount in each row of an Excel Table.

#### **Excel Table column formula notes:**

- 1) The Excel Table feature automatically copies them down the column.
- 2) They are Single Input-Output Formulas.
- 3) If you refer to a cell that needs to be locked, you must use the F4 key to lock the cell.
- 4) Dynamic Spilled Array Formulas (DSARs) are not allowed in Excel Table columns because the auto-copy feature of the Excel Table is in conflict with the auto-spilling of the DSARs.

## 19) Examples formulas by data type

A	B	C	D	E	F	G	H	I
2	<b>Formulas by Data Type:</b>							
3								
4	<b>Number Formulas</b> deliver numbers results (numbers, times, dates)							
5	Units Sold	43						
6	Price per Unit	22.50						
7	Total Sales Amount = Price * Units	967.5	=C6*C5					
8								
9	<b>Text Formulas</b> deliver text results							
10	Text: Description from Invoice	Purchased: CSQuad-4396						
11	Lookup Code for Product	4396	=TEXTAFTER(C10,"-")					
12	Convert text number to number (any math operation)	4396	=TEXTAFTER(C10,"-")+0					
13	Lookup Price	39.95	=XLOOKUP(--TEXTAFTER(C10,"-"),G9:G11,I9:I11)					
14								
15	<b>Logical Formulas</b> deliver logical values, or Boolean values (TRUE or a FALSE)							
16	Is text number equal to number, number?	FALSE	=C11=C12					

## 20) Convert text numbers to numbers

- i. You can convert any text number (number, time, date) back to a number with any math operation, such as adding zero or using a double negative.
- ii. Or the functions:
  1. VALUE to convert a text number to a number
  2. DATEVALUE to convert a text date to a serial number date
  3. TIMEVALUE to convert a text time to a serial number time

A	B	C	D	E	F
2	<b>Convert any text number (number, time, date) back to a number with any math operation or a functions.</b>				
4	Text Number	43			
5	Number	43	=C4+0		
6	Text Number	43			
7	Number	43	=VALUE(C6)		
8	Text Date	12/27/2024			
9	Serial Number Date	45653	=C8+0		
10	Text Date	12/27/2024			
11	Serial Number Date	45653	=DATEVALUE(C10)		
12	Text Time	10:25 PM			
13	Serial Number Time	0.934027778	=C12+0		
14	Text Time	10:25 PM			
15	Serial Number Time	0.934027778	=TIMEVALUE(C14)		

21) Examples of types of formulas by calculation type

- i. **Single Input-Output Formula:** Use when you must work on a computer without MS 365 Excel, or you must send the workbook to someone who did not have MS 365 Excel.
- ii. **Dynamic Spilled Array Formula:** Use when you are working on a computer with MS 365 Excel and you will not have to add new records to the table later. Advantages over Single Input-Output Formulas: 1) Do not have to lock cell references, 2) Do not have to manually copy the formula, 3) When editing formula, you do not have to manually re-copy the formula because it spills.
- iii. **Excel Table Formulas:** Use them in any version of Excel when you want to add or remove records to the table later and you want the formula to automatically copy down. Also use them when the table is going to be used in Power Query or Power Pivot (because they require them).
- iv. **Aggregate Formulas:** Use when you want to aggregate.
- v. **Scalar Array Formulas:** Use them when you do not want all the details of a spilled formula, but you would like to use an array formula result in an aggregate function like SUM.


Goal: Calculate tax for each sales amount, then add												
<b>Single Input-Output Formulas</b>			<b>Dynamic Spilled Array Formulas</b>			<b>Excel Table Column Formulas</b>			<b>Excel Table Formula</b>			
<b>Tax Rate</b>	0.0975		<b>Tax Rate</b>	0.0975		<b>Tax Rate</b>	0.0975		<b>Total Tax From Excel Table</b>			
									2.04			
<b>Sales</b>	<b>Tax Amount</b>		<b>Sales</b>	<b>Tax Amount</b>		<b>Sales</b>	<b>Tax Amount</b>		K7: =SUM(TaxTable[Tax Amount])			
4.49	\$0.44		4.49	\$0.44		4.49	0.44					
3.43	\$0.33		3.43	\$0.33		3.43	0.33					
4.98	\$0.49		4.98	\$0.49		4.98	0.49		<b>Scalar Array Formula</b>			
4.49	\$0.44		4.49	\$0.44		4.49	0.44					
3.44	\$0.34		3.44	\$0.34		3.44	0.34		<b>Total Tax calculated in single cell</b>			
<b>Total</b>	<b>\$2.04</b>		<b>Total</b>	<b>\$2.04</b>		<b>Total</b>	<b>2.04</b>		2.04			
			<b>Aggregate Formulas</b>						K14: =SUM(ROUND(B9:B13*C6,2))			
C9: =ROUND(B9*\$C\$6,2)			F9: =ROUND(E9:E13*F6,2)			I9: =ROUND([[@Sales]*\$I\$6,2)			<b>Incorrect:</b>			
C14: =SUM(C9:C13)			F14: =SUM(F9#)			I14: =SUBTOTAL(109,[Tax Amount])			<b>Each Amount Must Be Rounded!</b>			
									2.03			
<b>Single Input-Output Formulas Notes:</b>			<b>Dynamic Spilled Array Formulas Notes:</b>			<b>Excel Table Column Formula Notes:</b>			K19: =ROUND(SUM(B9:B13)*C6,2)			
Use if you have to send a solution to someone without M365 Excel.			Use because of these advantages: 1) Did not need to lock cell references. 2) Did not need to manually copy the formula 3) Editing a formula is faster and easier.			Use when you will add new records to your table. You still must lock references, but formulas are automatically copied down column.						

## 22) Excel Table feature

- i. We use Excel Tables so that when we add new rows or columns to our Excel Table, the formulas and PivotTables and other objects that are pointing to the ranges in the Excel Table will automatically update.
- ii. Requirements for an Excel Table:
  1. You must have a Proper Data Set (Field Names in 1st Row, 2) Records in Row, 3) Empty Cells All the way around)
  2. Click in one cell in Proper Data Set
  3. Click on Table button in Table group in Insert Ribbon Tab (or just use keyboard: Ctrl + T)
  4. Here is a picture of what we did in class:

Excel Table Formulas	
Tax Rate	0.1075
Sales (\$)	Tax Amount (\$)
1049.99	
1765.21	
537.03	
860.03	
947.61	
Total (\$)	

5. Name Table in Table Design Ribbon Tab, Properties group. You can NOT use spaces in the Table Name.



- iii. Add new records to Excel Table by typing or pasting new data in the first row below the Excel Table.
- iv. If new rows and columns are not added to Excel Table, you will need to change option settings for Excel Tables: File menu, Options button, Proofing tab on left, "AutoCorrect Options" button, check "Include new rows and columns in table".

23) Dynamic Spilled Array Formula to Build Budget

	A	B	C	D	E	F	G	H	I	J	K
1											
2			Dynamic Spilled Array Formulas to build a budget								
3											
4				<b>Jan</b>	<b>Feb</b>	<b>Mar</b>	<b>Apr</b>	<b>May</b>	<b>Jun</b>	<b>Total</b>	
5			<b>Sales:</b>	21,730	21,668	27,376	29,647	30,163	35,269	165,853	
6			<b>Expense as a % of Sales</b>								
7		0.375	COGS	8,149	8,126	10,266	11,118	11,311	13,226	62,195	J5: =SUM(D5:I5)
8		0.135	Salaries and Wages	2,934	2,925	3,696	4,002	4,072	4,761	22,390	D7: =ROUND(D5:I5*B7:B19,2)
9		0.075	Rent	1,630	1,625	2,053	2,224	2,262	2,645	12,439	J7: =SUM(D7:I7)
10		0.015	Supplies	326	325	411	445	452	529	2,488	D20: =SUM(D7:D19)
11		0.0475	Depreciation	1,032	1,029	1,300	1,408	1,433	1,675	7,878	D21: =D5:J5-D20:J20
12		0.035	Utilities	761	758	958	1,038	1,056	1,234	5,805	
13		0.0375	Advertising	815	813	1,027	1,112	1,131	1,323	6,219	
14		0.025	Administration	543	542	684	741	754	882	4,146	
15		0.015	Phone and Internet	326	325	411	445	452	529	2,488	
16		0.01	Repairs & Maintenance	217	217	274	296	302	353	1,659	
17		0.01	Travel	217	217	274	296	302	353	1,659	
18		0.025	Interest	543	542	684	741	754	882	4,146	
19		0.0875	Tax	1,901	1,896	2,395	2,594	2,639	3,086	14,512	
20			<b>Total Expenses</b>	<b>19,394</b>	<b>19,339</b>	<b>24,433</b>	<b>26,460</b>	<b>26,921</b>	<b>31,478</b>	<b>148,024</b>	
21			<b>Net Income</b>	<b>2,336</b>	<b>2,329</b>	<b>2,943</b>	<b>3,187</b>	<b>3,242</b>	<b>3,791</b>	<b>17,829</b>	

24) Rounding numbers.

i. ROUND Function

1. The ROUND function rounds a number to a specified number of digits and has these arguments:  
ROUND(number, num\_digits)
2. Typical situations where you need to use ROUND are payroll, invoicing, taxes.
3. The ROUND function must be used in situations where:
  - You are required to round (like with Money)
  - There are extraneous decimals
  - The amounts will be used in subsequent formulas
4. Examples:

	A	B	C	D	E	F	G	H
2	<b>ROUND function</b>							
3		Sales (\$)	Round to nearest penny					
4		43.687	43.69	C4: =ROUND(B4:B5,2)				
5		43.684	43.68					
7		Sales (\$)	Round to nearest dollar					
8		43.687	44	C8: =ROUND(B8:B9,0)				
9		43.490	43					
11		% Grade	Round to nearest tenth of a %					
12		85.274%	85.300%	C12: =ROUND(B12,3)				
14	Note: you must look at underlying number when deciding what position to round too (formulas do not 'see' Number Formatting). The underlying number is: 0.85274							

ii. MROUND function

1. The MROUND function returns a number rounded to the desired multiple and has these arguments:  
MROUND(number, multiple)
2. Typical situations where you need to use MROUND are payroll and pricing
3. Examples:

	A	B	C	D	E	F
16	<b>MROUND function</b>					
17		Nearest:	\$5			
19		Price	Rounded to nearest \$5			
20		\$43.687	45	C20: =MROUND(B20,\$C\$17)		
21		\$42.490	40			
23		Nearest 15 Min	00:15:00			
25		Time Value	Rounded to nearest 00:15 min			
26		8:06 AM	08:00:00 AM	C26: =MROUND(B26,\$C\$23)		
27		8:10 AM	08:15:00 AM			

- iii. CEILING.MATH function
  1. The CEILING.MATH function rounds a number **UP** to the nearest integer or to the nearest multiple of significance. and has these arguments: CEILING.MATH(number, [significance], [mode])
  2. A typical situations where you need to use is in pricing
- iv. FLOOR.MATH function
  1. The FLOOR.MATH function rounds a number **DOWN** to the nearest integer or to the nearest multiple of significance. and has these arguments: FLOOR.MATH(number, [significance], [mode])
  2. A typical situations where you need to use is in pricing
- v. Examples of CEILING.MATH and FLOOR.MATH:

	A	B	C	D	E	F
28						
29	<b>CEILING.MATH and FLOOR.MATH functions</b>					
30	<b>Nearest 15 Min</b>		00:15:00			
31						
32	<b>Time Value</b>		<b>Rounded to nearest 00:15 min</b>			
33		8:06 AM	08:15:00 AM	C33: =CEILING.MATH(B33:B34,C30)		
34		8:10 AM	08:15:00 AM			
35						
36	<b>Down to nearest</b>		\$5			
37						
38	<b>Price</b>		<b>Rounded down to nearest \$5</b>			
39		\$44.990	40	C39: =FLOOR.MATH(B39:B40,C36)		
40		\$40.010	40			



## 25) Date and Time Number Formatting and formulas

### i. Dates

1. Serial number date values use Number Formatting to display dates, but under the Number Formatting is a count of the number of days since Dec 31, 1899.
2. Examples: 1/1/1900 = 1, 12/30/2024 = 45656.
3. This allows date calculations with formulas, such as the date logic:
  - $\text{EndDate} - \text{BeginDate} = \text{number of days between the two dates.}$





A	B	C	D	E	F	G	H	I
2	<b>Goal:</b> Calculate number of days that the invoice is past due							
4	<b>Invoice Due Date</b>	10/25/2024						
5	<b>Today</b>	11/20/2024						
6	<b>Days Past Due</b>	26	=C5-C4					
8	<b>Goal:</b> Calculate number of workdays (Fri. & Sat. non-workdays)							
10	<b>Start</b>	<b>End</b>	<b>Days:</b>					<b>Holidays</b>
11	1/21/2024	3/14/2024	38	=NETWORKDAYS.INTL(B11,C11,7,I11:I13)				Mon, 1/15/24
12								Mon, 3/4/24
13								Tue, 3/5/24

### ii. Times

1. Serial number time values use Number Formatting to display times, but under the Number Formatting is a number that represents the proportion of a 24 hour day.
2. Example: 8 AM =>  $8/24 = 1/3 = 0.333333333333333$ .
3. This allows time calculations with formulas, such as the time logic:
  - $(\text{EndTime} - \text{BeginTime}) * 24 = \text{hours between two times.}$

A	B	C	D	E	F	G
14	<b>Goal:</b> Calculate the rounded gross pay amount					
16	<b>Hourly Wage (\$)</b>	<b>Start Time</b>	<b>End Time</b>	<b>Time Value</b>	<b>Hours Worked</b>	<b>Gross Pay (\$)</b>
17	27.55	8:00 AM	1:05 PM	5:05 AM	5.0833333	140.05
18				=D17-C17	=E17*24	=ROUND(F17*B17,2)
20	<b>Goal:</b> Round time value up to nearest 15 minutes and then calculate the rounded gross pay amount					
22	<b>Hourly Wage (\$)</b>	<b>Start Time</b>	<b>End Time</b>	<b>Time Value</b>	<b>Hours Worked</b>	<b>Gross Pay (\$)</b>
23	27.55	8:00 AM	1:05 PM	5:15 AM	5.25	144.64
24				=CEILING.MATH(D23-C23,"00:15")		
25					=E23*24	=ROUND(F23*B23,2)

## 26) Modes of cell editing

<b>Mode of cell when you are creating or editing a Worksheet Formula:</b>	
* The mode of a cell is listed in the lower-left corner of the status bar.	
* If you want to toggle between the modes, you can press the F2 key.	
* The four modes the status bar lists tell you what you can do with a formula:	
	1) <b>Ready</b> indicates that the cell is selected and ready for you to enter text, a number, or a formula.
	2) <b>Enter</b> indicates that you can use arrow keys or the mouse to select a range or the keyboard to enter a formula, or to type formula elements.
	3) <b>Point</b> indicates that that you are using your arrow keys or mouse to select a range. You can type formula elements in this mode also.
	4) <b>Edit</b> indicates that that the arrow keys will move the insertion point cursor from left to right in the formula.

27) Logical tests in Excel worksheet formulas

- i. The figure below shows examples for different types of logical tests and shows the different formula elements and functions that are used with logical formulas.
- ii. The SUMIFS, COUNTIFS, AVERAGEIFS and other similar IFS functions do AND Logical Tests by default.

A	B	C	D	E	F	G	H	I	J	K	L
2	<b>Type of Logical Tests:</b>	<b>Condition 1</b>	<b>Condition 2</b>	<b>Task</b>	<b>Result</b>	<b>Formula</b>	<b>Date</b>	<b>SalesRep</b>	<b>Product</b>	<b>Sales (\$)</b>	
3	Single Condition	Quad		Ave. Sales	533.3333	=AVERAGEIFS(L3:L23,K3:K23,C3)	1/1/25	Luong	Quad	100	
4	NOT Logical Test (not equal)	<>Chantel		Count Sales	15	=COUNTIFS(J3:J23,C4)	1/1/25	Sioux	Aspen	200	
5	OR Logical Test (1 or more TRUE over 1 column)	Luong	Sioux	Add Sales for L OR S	4300	=SUM(SUMIFS(L3:L23,J3:J23,C5:D5))	1/3/25	Bree	Aspen	900	
6	OR Logical Test (1 or more TRUE over 2 columns)	Bree	Quad	Count sales for B OR Q	7	=SUM(IF((J3:J23=C6)+(K3:K23=D6),1))	1/2/25	Chantel	Aspen	100	
7	AND Logical Test (All TRUE)	Bree	700	Count Bree Sales Bigger Than 700	2	=COUNTIFS(J3:J23,C7,L3:L23,">"&D7)	1/2/25	Luong	Aspen	300	
8	More...						1/2/25	Luong	Aspen	400	
9							1/1/25	Chantel	Aspen	100	
10	<b>Logical functions:</b>	<b>Condition 1</b>	<b>Condition 2</b>	<b>Task</b>	<b>Result</b>	<b>Formula</b>	1/3/25	Bree	Quad	800	
11	ISNUMBER, ISNA and other IS functions	43		Is the value a number?	FALSE	=ISNUMBER(C11)	1/2/25	Sioux	Aspen	200	
12	AND, OR and other aggregate logical functions	0	0	Are both greater than zero?	TRUE	=AND(C12>0,D12>0)	1/2/25	Sioux	Aspen	500	
13	IF and IFS function	Bree	Quad	Add sales for B OR Q	4100	=SUM(IF((J3:J23=C6)+(K3:K23=D6),L3:L23))	1/1/25	Sioux	Aspen	200	
14	FILTER and SWITCH	Bree	Quad	Add sales for B OR Q	4100	=SUM(FILTER(L3:L23,(J3:J23=C6)+(K3:K23=D6)))	1/1/25	Luong	Quad	500	
15	SUMIFS, MINIFS and other IFS functions	Bree		Minimum sale by B	700	=MINIFS(L3:L23,J3:J23,C15)	1/1/25	Sioux	Aspen	300	
16							1/2/25	Chantel	Aspen	500	
17	<b>Comparative Operators:</b>	<b>Three ways to use Comparative Operators:</b>									
18	Equal: =	1) Put condition in cell, like: <>Chantel									
19	Not Equal: <>	2) Direct logical test like: K3:K23=D6 or D12>0									
20	Greater than: >	3) Join operator to condition, like: ">"&E7									
21	Greater than or equal to: >=						1/2/25	Luong	Aspen	800	
22	Less than: <						1/1/25	Chantel	Quad	400	
23	Less than or equal to: <=						1/2/25	Sioux	Aspen	400	
							1/3/25	Bree	Quad	700	
							1/2/25	Chantel	Aspen	500	
							1/1/25	Luong	Aspen	400	

28) Lookup functions

i. Examples of some of the worksheet lookup functions:

A	B	C	D	E	F	G	H	I	J	K	L
2	<b>XLOOKUP to lookup a record:</b>						<b>EmployeeID</b>	<b>First</b>	<b>Last</b>	<b>StartDate</b>	<b>Department</b>
3							4369-9084	Sioux	Chin	10/6/2012	Accounting
4	<b>EmployeeID</b>	<b>First</b>	<b>Last</b>	<b>StartDate</b>	<b>Department</b>		4369-3979	Kip	Hensel	10/6/2020	Maintenance
5	4369-9084	Sioux	Chin	10/6/12	Accounting		4369-4774	Ty	Smithe	11/15/2011	Finance
6		=XLOOKUP(B5,H3:H5,I3:L5)									
7											
8	<b>LOOKUP to lookup tax from sorted A-Z tax table:</b>						<b>Taxable Amount</b>	<b>Tax Rate</b>			
9							0	2.00%			
10	<b>Taxable Amount</b>	5500					950	3.50%			
11	<b>Tax Rate</b>	4.75%					2750	4.75%			
12		=LOOKUP(C10,H9:I12)						6500	6.50%		
13											
14	<b>FILTER to use one lookup value to return multiple items:</b>						<b>Date</b>	<b>Product</b>	<b>Sales (\$)</b>		
15							1/1/25	Quad	100		
16	<b>Product</b>		<b>Date</b>	<b>Product</b>	<b>Sales (\$)</b>		1/1/25	Aspen	200		
17	Quad		1/1/25	Quad	100		1/2/25	Aspen	400		
18			1/3/25	Quad	800		1/1/25	Aspen	100		
19							1/3/25	Quad	800		
20											
21	<b>XMATCH to compare two lists:</b>										
22											
23	<b>List 1</b>		<b>List 2</b>	<b>In both?</b>	<b>Just in List 2?</b>	<b>Names just in List 2?</b>	<b>Names in both?</b>				
24	Chantel		Miki	FALSE	TRUE	Miki	Chantel				
25	Sheliadawn		Chantel	TRUE	FALSE	Gigi	Luong				
26	Luong		Gigi	FALSE	TRUE	Sioux					
27	Frank		Sioux	FALSE	TRUE	Pops					
28	Bree		Luong	TRUE	FALSE						
29	Bobbi		Pops	FALSE	TRUE						

29) Formula Errors

<b>Error Message</b>	<b>What It Means</b>
#DIV/0!	A divide-by-zero error has occurred.
#REF!	The formula is using a cell reference that has been deleted or an otherwise invalid cell reference.
#NAME?	An Excel built-in function or defined name is misspelled or word data in a formula is not in double quotation marks.
#N/A	A not available error has occurred, such as when XLOOKUP or XMATCH functions can't find a match.
#VALUE!	An invalid operator or argument (such as ="Red"*12 or =ROUND(G43,"Two").
#NULL!	There is no intersection for a space operator lookup.
#NUM!	The number is too big or small (and not between $-1*10^{307}$ and $1*10^{307}$ ), a formula or function contains invalid numeric values, or an iterative function such as IRR cannot find an answer.
#####	The column is not wide enough to display data (values) or a date or time is a negative value.
Circular cell reference	A formula contains a reference to the cell the formula sits in. This would occur, for example, if =SUM(A1:A3) were in cell A1. The formula doesn't know what to do because it is looking at itself.
#SPILL!	Something in a cell is blocking a dynamic spilled array formula from spilling the full array of values.
#CALC!	An array formula cannot make a calculation, such as when the formula tries to calculate an array within an array, which is not allowed.
#BUSY!	The data type dot operator is waiting to get an answer, as with a stock price for a Stocks data type or a population number for a Location data type.